Use Z-transform to solve a recurrence

```
> restart:
```

Check working using Fibonacci

```
> rsolve( { F(n+1)=F(n)+F(n-1), F(0)=1,F(1)=1}, F ) ;
```

$$\left(\frac{1}{10}\sqrt{5}+\frac{1}{2}\right)\left(\frac{1}{2}\sqrt{5}+\frac{1}{2}\right)^n+\left(\frac{1}{2}-\frac{1}{10}\sqrt{5}\right)\left(-\frac{1}{2}\sqrt{5}+\frac{1}{2}\right)^n \quad (1)$$

Use Z-tranform to compute 1+2^2+3^2+...+n^2

```
> REC := S(n+1)=S(n)+(n+1)^2 ;
  INI := S(0)=0,S(1)=1 ;
```

$$REC := S(n+1) = S(n) + (n+1)^2$$

$$INI := S(0) = 0, S(1) = 1 \quad (2)$$

```
> rsolve( { REC, INI}, S ) ; factor(simplify(%)) ;
```

$$1-3\,(n+1)\left(\frac{1}{2}\,n+1\right)+n+2\,(n+1)\left(\frac{1}{2}\,n+1\right)\left(\frac{1}{3}\,n+1\right)$$

$$\frac{1}{6}\,n\,(n+1)\,(2\,n+1) \quad (3)$$

Step 1, perform Z-tranform

```
> ztrans( REC, n, z ) ;
  Z_REC := subs( ztrans(S(n), n, z)=S(z), %) ;
```

$$z\,ztrans(S(n),n,z) - S(0)\,z = ztrans(S(n),n,z) + \frac{z\,(z+1)}{(z-1)^3} + \frac{2\,z}{(z-1)^2} + \frac{z}{z-1}$$

$$Z\_REC := z\,S(z) - S(0)\,z = S(z) + \frac{z\,(z+1)}{(z-1)^3} + \frac{2\,z}{(z-1)^2} + \frac{z}{z-1} \quad (4)$$

Apply the initial conditions

```
> Z_REC_WITH_INI := subs( INI, Z_REC) ;
```

$$Z\_REC\_WITH\_INI := z\,S(z) = S(z) + \frac{z\,(z+1)}{(z-1)^3} + \frac{2\,z}{(z-1)^2} + \frac{z}{z-1} \quad (5)$$

Solve the recurrence in z space

```
> SOL_Z := op( solve( Z_REC_WITH_INI, {S(z)} ));
```

$$SOL\_Z := S(z) = \frac{z^2\,(z+1)}{(z-1)^4} \quad (6)$$

Partial fraction expansion of S(z)/z as

```
> PFE := A1/(z-1)+A2/(z-1)^2+A3/(z-1)^3+A4/(z-1)^4 ;
```

$$PFE := \frac{A1}{z-1} + \frac{A2}{(z-1)^2} + \frac{A3}{(z-1)^3} + \frac{A4}{(z-1)^4} \quad (7)$$

Compute A4

```
> Sz_over_z := rhs(SOL_Z)/z ;
```

$$Sz\_over\_z := \frac{z\,(z+1)}{(z-1)^4} \quad (8)$$

```
> COEFFS := A4 = subs( z=1, Sz_over_z*(z-1)^4),
```

```
            A3 = subs( z=1, diff(Sz_over_z*(z-1)^4,z) ),
            A2 = subs( z=1, diff(Sz_over_z*(z-1)^4,z,z)/2 ),
            A1 = subs( z=1, diff(Sz_over_z*(z-1)^4,z,z,z)/6 ) ;
```

$$COEFFS := A4 = 2, A3 = 3, A2 = 1, A1 = 0 \quad (9)$$

```
> simplify(subs(COEFFS, PFE*z)) ;
```

$$\frac{z^2 (z+1)}{(z-1)^4} \quad (10)$$

Invert each term of Z-transform

```
> B0 := 1 ;              # binomial(n,0)
  B1 := n ;              # binomial(n,1)
  B2 := n*(n-1)/2 ;      # binomial(n,2)
  B3 := n*(n-1)*(n-2)/6 ; # binomial(n,3)
```

$$B0 := 1$$

$$B1 := n$$

$$B2 := \frac{1}{2} n (n-1)$$

$$B3 := \frac{1}{6} n (n-1) (n-2) \quad (11)$$

```
> SOL := factor(expand(subs(COEFFS,A1*B0+A2*B1+A3*B2+A4*B3))) ;
```

$$SOL := \frac{1}{6} n (n+1) (2 n+1) \quad (12)$$

Invert the Z-transform using MAPLE

```
> factor(invztrans( SOL_Z, z, n )) ;
```

$$invztrans(S(z), z, n) = \frac{1}{6} n (n+1) (2 n+1) \quad (13)$$

Sum of first n-number 1+2+3+...+n

```
> factor(simplify(rsolve( { S(n+1)=S(n)+n+1, S(0)=0,S(1)=1}, S ))) ;
```

$$\frac{1}{2} n (n+1) \quad (14)$$

Sum of first n-cube 1^3+2^3+3^3+...+n^3

```
> factor(simplify(rsolve( { S(n+1)=S(n)+(n+1)^3, S(0)=0,S(1)=1}, S ))
  ) ;
```

$$\frac{1}{4} n^2 (n+1)^2 \quad (15)$$